
1 Executive Summary

This report addresses the construction of carbon fibre wing boxes and the problems as-
sociated with using carbon fibre sheets rather than individual carbon fibre tapes. In the
case that the wing boxes are developable surfaces the lay up of carbon fibre sheets is
straightforward, since the fibres can follow the contours of the surface without any need
for shearing or extension of the fibres. To further expand the potential design space for
the wing boxes, this report investigates the lay up of sheets over non-developable surfaces
where some shearing of the sheet is required to achieve the desired results. In this report,
three analytical approaches are considered, driven by the results from numerical studies
on different surface geometries. Each of the approaches offers insights as to the type of
geometric perturbations achievable when constrained by a maximum shear angle.

Figure 1: A sketch of a plane wing, showing the location of the wing box within in. The upper
and lower components of the wing box are coloured black and blue, while the fore and aft
extensions to the wing are shown in red.

2 Problem Description

Within the aerospace industry there is an ongoing trend to manufacture large parts of new
aircraft using carbon fibre rather than aluminium/titanium as is more conventional. This
trend presents challenges; there is considerable expertise available within the manufac-
turing community to produce a wide variety of complex geometries by machining aluminium
— such expertise is not yet readily available for carbon-fibre parts which are constructed
from flexible, but inelastic, fabric sheets.

Current wing box designs do not take into account the difficulties associated with carbon-
fibre manufacturing methods. In particular, they do not consider that they should be easily
constructed from flexible sheets that cannot be stretched. Consequently, the manufacture
of existing wing-box designs from carbon fibre sheets is time consuming and difficult.

The problem considered here is to characterise the design constraints for carbon-fibre
wing boxes, in other words, to find which designs are straightforward to manufacture.

With carbon-fibre construction, the wing box is constructed as two separate components
fastened together. Each part contains a wing surface (upper or lower) and a vertical spar
that spans the cross-section of the wing. See figure 1.

We consider that the wing surface profile is given, with a set curvature both in the span-
wise direction, and the fore-aft direction. The typical wing is 14m meters in length and has
a radius of curvature of the order of 65m. Moreover, the wing has a slope of 15◦ near the
aircraft, 7◦ at a distance of 4m from the aircraft and 3◦ at the wing tip.

Figure 2: The simplest under-and-over carbon-fibre weave formed by ribbons comprising
many individual fibres, initially crossing at right angles (left). Other weave patterns and
initial fibre angles are also available. When subjected to simple shear, the gaps between
the ribbons are reduced. The critical shear occurs when then ribbons come into contact
(right).

A spar then joins to the wing cover, roughly at right-angles to the wing surface, with a
height of the order of 1m.

The carbon fibre fabric comprises a woven mesh of individual fibres. Typically the fibres
are grouped together to form flat ribbons, which are then woven in some pattern. These
mesh fabrics do not stretch but can be sheared freely up to a critical shear angle (about 7◦
for the fabrics considered). The critical shear occurs when the ribbons come into contact,
and the weave then prevents further shearing. See figure 2. In the construction process,
several layers of carbon fibre fabric much be laid over a mould of the wing and spar shape.
For structural strength, each single sheet should smoothly cover the wing and spar, without
needing cuts or leaving wrinkles.

The Gaussian curvature of the wing surface is sufficiently small to allow the carbon-fibre
sheets to be laid smoothly over the surface. Laying the carbon-fibre sheets over the edge
and onto the spar presents greater difficulties. Airbus already appreciate that if two devel-
opable surfaces are joined by a so-called ‘origami fold’ (i.e. one in which the two surfaces
could be flattened out into a single developable sheet) then the carbon fibre sheets can
be applied without any problems. When attempting to construct wing box components with
a deviation from such a fold, it has been found that is is difficult to get the sheets to lie
smoothly over the surfaces without rucks appearing.

The design space of a wing box that comprises an origami fold is too small to be useful.
Instead, the main problem is to determine if one can modify the curvature of the spar by
some amount and still be able to cover the wing cover and spar smoothly with carbon fibre,
only allowing shear deformation of the fabric and with a shear angle below a given bound.

The Study Group was therefore asked to investigate the laying of carbon fibre sheets over

different edge geometries, with a particular focus on trying to quantify how much deviation
is permitted from the ideal ‘origami fold’ given the critical shear angle of the fabric.

3 Model Problem

3.1 Wing and Spar Geometry

To simplify the problem we shall focus on the edge between the wing and the spar, and take
the surfaces either side to be developable. Furthermore, we shall consider the special case
where the generators are parallel within each surface, and that the two sets of generators
are perpendicular to each other.

This leads to a natural coordinate system, in which the x axis is parallel to the generators
on the wing cover (roughly fore–aft) and the z axis is parallel to the generators in the spar
(roughly vertical). The y axis is then oriented along the span of the wing. To describe the
shape of the wing and spar we just need to specify how the curvature of each surface
varies along the span of the wing.

We introduce two functions for the profiles of the two surfaces: gw(y) for the wing and gs(y)
for the spar. The surface of the wing is then given by

z = gw(y) for x > gs(y) , (1)

while surface of the spar is given by

x = gs(y) for z < gw(y) . (2)

An origami fold occurs precisely when the two surfaces have the same profiles, i.e.

gw(y) = gs(y). (3)

In this case, the intersection between the spar and the wing cover is a curve that lies on
a plane and the two surfaces can be covered by a single carbon-fibre sheet without any
deformation.

We wish to investigate non-origami folds, i.e. gw(y) 6= gs(y), and whether or not these can be
covered smoothly by a carbon-fibre sheet subject to a given critical shear constraint.

3.2 Carbon Fibre Model

We focus on a single sheet of fabric consisting of two sets of inter-woven ribbons, which
are initially at 90◦ to each other. The carbon fibre weave size is of the order of a few

millimetres, which is much smaller than the length scales of the wing cover and spar.

When it is deformed, the angle σ between ribbons changes and we define the shear angle
as the difference between that angle and a right angle:

θshear =
∣∣∣π
2
− σ

∣∣∣ (4)

We consider the fibre ribbons in the sheet as inextensible rods, hinged freely at the square
lattice of cross-over points. The ribbons are able to bend both within the plane of the
sheet (to allow for changes in the shear) and out of the plane (to allow for curvature in the
surfaces and at the edge).

Because the scale of the fibres and the weave is much smaller than the dimensions of the
wing, it is appropriate to use continuum approximation for the sheet if convenient.

For structural reasons, most of the carbon fibre sheets have fibres aligned along the length
of the wing, so we shall focus on this case in this report. We shall generally assume that
the fibre sheet is first laid undeformed on the developable wing surface and then folded
around to cover the spar, with shearing on the spar introduced as necessary.

3.3 Numerical Solution

To provide insights into the resulting geometry of carbon fibre sheets draped over arbi-
trary surfaces, a numerical solution method was developed. The underlying assumptions
embedded within the numerical method are as specified in sections 3.1 and 3.2.

The solution process begins with fixing the positions of two non-parallel fibres on the spec-
ified surface. With these fibres as boundaries and for given weave size, the placement of
all the remaining fibres within the carbon fibre sheet is entirely determined.

Nodes, representing the intersection of two fibres, are placed along each fibre with the
distance between them being the weave size δ. A complete grid of nodes is constructed
starting at the intersection of the two initial fibres; two adjacent nodes are chosen and a
point that is δ away from both nodes is found and a new node is placed there. See figure 3
for a schematic.

The problem of finding the location of a new node is a nonlinear root finding problem such
that

d(n1, ñ) = δ, d(n2, ñ) = δ

where n1 and n2 are the locations of two existing nodes, ñ is the location of the new node,
and d(·, ·) is the geodesic distance between two points on the specified surface. In the case

Figure 3: A schematic of the numerical construction of a grid of fibres.

where the surface is developable, the geodesic distance is obtained using the standard
formula for arclength. A complication arises in the setting considered here due to the
boundary between the surface of the wing and the surface of the spar; to obtain geodesic
distances in this case, an optimisation problem is solved whereby the position that the fibre
crosses the boundary is considered to be a parameter that can be varied to minimise the
overall distance between the nodes.

For the numerical codes used in this report, please refer to section 8.

3.4 Qualitative Features of a Typical Solution

A qualitative sketch of a typical solution is shown in figure 4. In this case we have used

gw(y) ≈ 0.2 y2 , gs(y) ≈ 0.4 y2 .

The upper part shows the wing, and the lower the spar. In both cases the developable
surfaces have been flattened out, so the horizontal coordinate is arc-length along each
surface rather than y.

The carbon-fibre sheet is placed over the wing section without deformation, leading to a
regular grid on the upper part of figure 4. The layout on the spar in the lower part is then
determined by fixing the left-most cross-wing ribbon (green) to be vertical, and by the fixed
positions of both sets of ribbons that are imposed at the edge (black) from the solution on
the wing.

There are two distinct regions of behaviour on the spar. Region I where along-wing rib-
bons (blue) originate from the left-hand boundary, and Region II where along-wing ribbons
originate from the wing edge.

In Region I, the cross-wing ribbons (green) are vertical, and the along-wing ribbons (blue) all
have the same shape, but are vertical translations of each other. The common shape of
the along-wing ribbons is caused by vertical shifts in the cross-wing ribbons. There are two

causes of these shifts. First the difference gw(y)− gs(y) at the corner gives a vertical shift
in the boundary condition at the top of the spar. Secondly, any deviation of the cross-wing
ribbons from vertical in Region II causes an additional upward displacement of the sections
lower down. We note that the gradient of the along-wing ribbons in Region I means that the
cross-wing ribbons lie closer together towards the right-hand side of the spar.

In Region II there is shearing of the cross-wing ribbons, to match between the upper bound-
ary at the wing and the lower boundary with Region I. At the upper boundary, the ribbons
are displaced horizontally because the different gradients of the wing and spar result dif-
ferent arc-lengths in the y-direction along the two surfaces. In this example, the spar has
larger gradient (g′s(y) > g′w(y)) and so has a longer arc-length in the y direction than the
wing. Thus the spacing between the cross-wing ribbons is increased here at the top of the
spar. At the lower boundary, the cross-wing ribbons are displaced horizontally because of
the slope of the along-wing-ribbons in Region I. Thus the spacing between the cross-wing
ribbons is decreased at the bottom of the spar.

We also note at this point that the solution in Region II is completely determined by the layout
of the carbon-fibre sheet on the wing. In Region I, however, we have some flexibility, from
the boundary condition imposed on the first cross-wing fibre at the left-hand side. In the
example in figure 4, we could have placed a small negative horizontal shear at the left-
hand side. This would have the effect of sheering the whole of Region II to the left. With
an appropriate shear on the left-hand side, we could reduce the maximum shear found in
Region II. This suggests it may be important how the sheet is laid down on the spar, and
which end is fixed first.

The solution also points to a potential ill-posedness issue if an along-wing ribbon starts on
the wing, goes over the edge onto the spar, and later crosses back over the edge and on
to the wing. In this case, laying out the wing without deformation would impose boundary
conditions at both ends of the section of the ribbon laying over the spar, leading to an ill-
posed problem. One would need to include some shearing on the wing to accommodate
this.

4 Perturbation of the origami solution.

In this section we consider a spar with a profile deviating from the origami solution. We
describe the modified profile of the spar with

gs(y) = gw(y) + f(y). (5)

instead of (3), where f(y) is the amount of deformation. This is illustrated on figures 5 and 6.

Figure 5: Wing with deformed spar. The blue curve illustrates how the spar can deviate
from the origami solution profile

Figure 6: Spar profile: origami spar (black) and modified spar (blue).

We consider first the simplest origami solutions, i.e. without perturbation, where the coor-
dinates of the carbon fibre nodes on the spar belonging to the same horizontal ribbon are
all at the same level. We consider a single row at time and set z = 0 for that row. Labelling
the coordinates of the ith junction along the row as (xi, yi, zi), we thus have

zi = 0, xi = gs(yi) = gw(yi) = gi. (6)

The yi are determined by the constraint that the distance between node is equal to the
mesh size λ. By Pythagoras’ theorem, we thus have

λ2 = (xi+1 − xi)
2 + (yi+1 − yi)

2 + (zi+1 − zi)
2. (7)

Introducing the horizontal spacing li = yi+1 − yi, we have

λ2 = (gi+1 − gi)
2 + l2i . (8)

Using the Taylor series of gw(y), and writing g′i = g′w(yi), we have

gi+1 ≈ gi + g′ili (9)

and so λ2 = (g′ili)
2 + l2i leading to

li =
λ

1 + g′2i
(10)

for the origami solution.

If we modify the shape of the spar, with a perturbation f(y), we then have the new coordi-
nates of the ith junction as (x̂i, ŷi, ẑi) where

x̂i = gw(ŷi) + fi(ŷi) , ŷi = yi +∆yi , ẑi = −fi(ŷi) , (11)

where ∆yi is the horizontal perturbation from the origami solution.

After the deformation, the distance between nodes must remain equal to the mesh size λ,
so

λ2 = (x̂i+1 − x̂i)
2 + (ŷi+1 − ŷi)

2 + (ẑi+1 − ẑi)
2 (12)

=
(
gi+1 − gi + fi+1 − fi + (g′i+1 + f ′i+1)∆yi+1 − (g′i + f ′i)∆yi

)2
+
(
yi+1 − yi +∆yi+1 −∆yi

)2
+
(
fi+1 − fi + f ′i+1∆yi+1 − f ′i∆yi

)2
. (13)

Defining
Si = ∆yi+1 −∆yi , (14)

and using (9) we have

gi+1 − gi + g′i+1∆yi+1 − g′i∆yi = g′ili + g′i+1(∆yi + Si)− g′i∆yi

= g′i(li + Si) + g′′i li(∆yi + Si). (15)

In the design of the wing, g′(y) is relatively small as g′(y) ≈ 0.25 for 15◦. Moreover, the radius
of curvature of the wing is much larger than the mesh size of the fibre and so λg′′ << 1. As
a result g′′i li << g′i and we can neglect the last term in (15). Similarly for f we have

fi+1 − fi + f ′i+1∆yi+1 − f ′i∆yi ≈ f ′i(li + Si). (16)

Combining (13), (15) and (16) we can write

λ2 ≈ (g′i(li + Si))
2 + 2(f ′i(li + Si))

2 + 2g′if
′
i(li + Si)

2 + l2i + S2
i + 2lisi.

Then, using (10) we have

0 =g′i(S
2
i + 2liSi) + 2(f ′2i + g′if

′
i)(l

2
i + S2

i + 2liSi) + S2
i + 2liSi

=(S2
i + 2Sili)(1 + (g′i + f ′i)

2 + f ′2i ) + 2l2i (f
′2
i + f ′ig

′
i) (17)

and so

Si =− li + li

√
1− 2(f ′2i + f ′ig

′
i)

1 + (g′i + f ′i)
2 + f ′2i

≈− li
f ′2i + f ′ig

′
i

1 + (g′i + f ′i)
2 + f ′2i

. (18)

We thus see that the lateral distance between the vertical ribbons decrease by Si which is
quadratic in the derivatives of f and g and can thus be neglected to first order approxima-
tion.

This means that as the shape of the spar is modified, the first order effect is that the
horizontal ribbons on the spar are simply shifted vertically by f(y) and this induces a shear
angle θ given by

tan(θ) =
zi+1 − zi
yi+1 − yi

=
f(yi+1)− f(yi)

yi+1 − yi

≈ df

dy
. (19)

The shear angle θmax is then given by

θmax = arctan

(
max

(
df

dy

))
. (20)

This is confirmed by the numerical solutions, except that there is some extra shear of the
fibre on the wing cover, but it is smaller than the shear on the spar. Hence our evaluation
of the maximum shear is a good estimate.

5 A Simple Special Solution

If the spar is not curved but is vertical, there exists a very simple and special solution which
can be constructed easily. For the spar we are considering, gs(y) = 0 everywhere. For this
solution the carbon fibres on the wing cover are parallel and orthogonal to the edge of the
(straight) spar. When they reach the spar, the strands must be pulled down vertically on
the spar as illustrated on figure (7)

1

3

2

5

4

6

a

Figure 7: Flat spar solution: front view of the spar. The red line marks the edge between
the spar and the wing cover.

If λ is the length of a mesh and if the first parallel strand is at a distance b from the spar
edge, then the first nodes on the spar, 1,2 and 5, are at a distance a = λ − b from the spar

edge. We can then chose point 3 to be vertically below point 1 and at a distance λ. Point
4 will then be vertically below point 2 and at a distance λ from it. This grantees that the
distance between points 3 and 4 is also λ. Point 6 can the be positioned just below point 5
and we see that on the spar the nodes will be aligned on a curve parallel everywhere to
the spar/wing interface.

We can then see that, as before, the maximum sheer is given by

θmax = arctan

(
max

(
df

dy

))
. (21)

The main problem with this solution is that it appears to be highly sensitive to perturbations.
It can be constructed numerically, but any small deviation leads to large shear on the wing
cover or the spar.

6 Alternative Analytic Treatment

6.1 Parallel ribbon segments

We begin by making the observation that on a developable surface, the assumption about
the ribbons being inextensible and pivoted at the crossing points means that each ‘cell’
bounded by four ribbon segments must be a parallelogram on the flattened surface. There-
fore following a row or column of cells, the transverse ribbon segments must all be parallel
to each other, i.e. have the same rotation.

In a manner reminiscent of characteristics, if the rotation angles of the segments are set at
the boundaries, then rows and columns of cells can be followed in to the interior in order
to determine the angles of the segments around each cell. This is depicted in figure 8. For
each cell, the difference in the rotation angles of the two pairs of ribbon segments gives
the shear angle of the cell.

6.2 Separation of scales

We make the assumption that |g′′s |, |g′′w| � λ−1, which says that the variation in the gradient
of the surfaces happens on a length scale much larger than the carbon-fibre mesh size.
This means that locally, on the mesh scale, the edge between the wing and spar appears
as a straight line to good approximation. The continuation of the mesh over the edge thus
corresponds to a simple fold, and this allows us to make progress in what follows below.

Formally, we need to ensure that there is negligible change in the angle that the edge

γ

β − α

β − α

Figure 8: In each row or column of cells on the spar, the transverse ribbons segments
must remain parallel. Therefore the rotation of the horizontal ribbon segments (blue) in
each column of cells is set by the conditions at the wing edge, while the rotation of the
vertical ribbon segments (green) in each row is set either by the left-hand boundary or by
the conditions at the wing edge, depending on where the horizontal ribbon originated.

makes on the wing and spar surfaces between adjacent ribbons. Let g′′max be the largest
absolute value of either second derivative, so that λg′′max � 1.

On the wing, cross-wise strands are at most a distance λ apart in the y coordinate. The
change in the gradient of the edge on either surface is then at most the product of the
longest possible distance, and the largest possible second-derivative: λg′′max which we have
assumed is small.

Shallow gradients in the wing and spar profiles can lead to span-wise ribbons being longer
distances apart when they meet the edge. But we can still show that the deviation in the
gradient between such ribbons much be small. Suppose that the difference between the
maximum and minimum gradients between two span-wise ribbons isD. Then there is a point
between them with a absolute gradient of at least D/2. Then since the rate of change of
the gradient is limited by g′′max, there is a length

` ≥ D

4g′′max

(22)

between the two ribbons, over which the modulus of the gradient is at least D/4. Over this
length, the g changes by

∆g ≥ `
D

4
≥ D2

16g′′max

. (23)

But we must have ∆g ≤ λ for the edge to be within the two ribbons. Hence

D2

16g′′max

≤ λ ⇒ D ≤ 4 (λg′′max)
1/2 � 1 . (24)

The fact that the maximum deviation in the derivatives g′s and g′w is small between adjacent
ribbons, means that the corresponding change in the angles α and β, as defined below in
26 and 27, must also be small. If these angles do not change much, then the edge is straight
to good approximation.

6.3 Local behaviour at the edge

We introduce arc-length coordinates sw and ss in the y direction on the wing and spar
respectively. These are related to the y coordinate in the usual way by

dsw
dy

=
(
1 + g′w

2
)1/2

,
dss
dy

=
(
1 + g′s

2
)1/2

. (25)

We now consider the mesh being folded over an approximately straight section of the edge
in the neighbourhood of the point (gs(y), y, gw(y)). The surfaces are shown in figure 9 in their
flattened-out forms. We define α as the angle that the edge makes on the wing surface,
relative to the (sw, z) coordinates. Similarly we define β for the spar, relative to the (ss, x)

coordinates. We then have

tanα =
dx

dsw

∣∣∣∣
edge

=
dgs
dy

dy

dsw
=

g′s(
1 + g′w

2
)1/2 , (26)

tanβ =
dz

dss

∣∣∣∣
edge

=
dgw
dy

dy

dsw
=

g′w(
1 + g′s

2
)1/2 . (27)

As shown in figure 9, the mesh is offset by an angle −α relative to a line normal to the edge
on the wing, and hence we have the same rotation relative to the normal line on the spar.
This normal line is rotated from the vertical by the angle β, and hence the net rotation of
the mesh on the spar is ψ = β − α.

Thus immediately below the edge at the top of the spar, the rotation angle of both sets of
ribbons is given by ψ = β − α. This angle can be computed via

tanψ = tan(β − α) =
tanβ − tanα

1 + tanα tanβ
=
g′w
(
1 + g′w

2)1/2 − g′s
(
1 + g′s

2)1/2(
1 + g′w

2
)1/2(

1 + g′s
2
)1/2

+ g′w g
′
s

. (28)

6.4 Shear angles from ribbon rotations

This rotation of the span-wise ribbons is propagated down the columns, while this rotation
of the cross-wise ribbons is propagated along the rows, as shown in figure 8. Additionally,
cross-wise ribbons starting from the left-hand boundary of the spar (rather than originating
on the wing) will have their rotation set by the boundary condition there. We denote the
rotation angle of the vertical ribbons set there by γ.

x = gs(y)

z = gw(y)

α

ψ = β − α

Figure 9: The rotation of the cells caused by a local straight fold at the wing–spar edge.
Then the angle of the edge on the wing is α and the angle of the edge on the spar is β, then
the rotation of the mesh due to the fold over the edge is β − α.

The shear angle experienced in each cell is then computed as the difference in the rotations
of the span-wise and cross-wise ribbons that form its boundary. In the region reached
by cross-wise ribbons originating on the wing, the maximum shear is given by the largest
difference in ψ values as y varies.

θmax,I = max
y,ŷ

∣∣ψ(y)− ψ(ŷ)
∣∣ = ψmax − ψmin , (29)

where ψmax and ψmin are the largest and smallest values of rotations ψ from along the edge.

In the region reached by cross-wise ribbons originating on the boundary, the maximum
shear is given by the largest difference between ψ and γ as y varies.

θmax,II = max
y

∣∣ψ(y)− γ
∣∣ (30)

This is minimised by taking γ = 1
2 (ψmax + ψmin), giving

θmax,II =
1

2
(ψmax − ψmin) . (31)

Moreover for any γ ∈ [ψmin, ψmax] we will have θmax,II ≤ θmax,I, so (29) provides the effective
constraint.

Insisting that θmax,I ≤ θ∗ is the maximum allowable shear angle in the carbon-fibre sheet,we
obtain the condition

ψmax − ψmin ≤ θ∗ , (32)

where ψ(y) is computed from gs(y) and gw(y) using (28).

7 A Continuum Model

As mentioned in Section 2, the radius of curvature of the wing is much greater than mesh
size and so it is reasonable to consider a continuum limit of infinitely small weave. The
woven ribbons lie on a given surface and correspond to a continuous deformation of a 2-
parameter family of curves in this surface z = z(x, y). If u and v are Lagrangian coordinates
on R2 measured along the fibres, the surface can be described as x = (x, y, z(x, y)) where x
and y are functions of u and v which satisfy the equations∣∣∣∣∂x∂u

∣∣∣∣ = 1,

∣∣∣∣∂x∂v
∣∣∣∣ = 1. (33)

These equations express the fact that the ribbons are inextensible and they can be written
explicitly in the form(

(34)

This model is the continuum limit of the discrete model described in Sections 4 and 6 and
also in [1].

Equations (34) are nonlinear and do not fall under the usual classification of partial differ-
ential equations. However it can be shown that u = constant and v = constant are character-
istics and thus prescribing x on a fibre u = constant and y on a fibre v = constant leads to a
well-defined solution. We will now present some examples.

Since the relative rotation of the fibres is restricted by the geometry of the woven fabric
it is necessary to add a constraint that the shear angle θ must be less than some limiting
value θmax and this leads to the condition

cos
(π
2
− θ
)
= sin(θ) =

∣∣∣∣∂x∂u · ∂y
∂v

∣∣∣∣ 6 sin(θmax). (35)

In order to make analytical progress we will now consider the solution of the pair of equa-
tions (34) when the surface z = z(x, y) only deviates from a plane by a small distance. This
will allow us to solve the equations using a regular perturbation procedure and will give
some useful insights. For most practical cases we will need to solve the problem numeri-
cally as is done below.

7.1 A smooth convex surface

We start by considering the woven material being draped over a smooth convex surface.
We assume that the surface is given locally by

z = −
√
ε (ax2 + 2bxy + cy2), where ac > b2 (36)

and ε is a small parameter. Then we can assume that the solution of (34) is a regular
perturbation of the flat solution in which the fabric is unsheared and write

z = 0, x = u+ εx1, y = v + εy1. (37)

In addition, we assume that x = 0 on u = 0 and y = 0 on v = 0 which determines the
orientation of the fabric over the surface. The solution to O(ε) is then

x = u− 2

3
ε(a2u3 + 3abu2v + 3b2uv2)

y = v − 2

3
ε(3b2u2v + 3bcuv2 + c2v3) (38)

and the shear can be determined using (35) as

sin(θ) = 2ε[abu2 + 2(ac− b2)uv + bcv2]. (39)

This analysis also holds in the limiting case when ac = b2 and the surface is locally cylindrical.
By taking b = c = 0, we see that there will be no shear in the fabric when it is draped over the
cylindrical shape z = −

√
εax2 as long as one of the weave directions lies along a generator

of the cylinder. On the other hand , if we take ac = b2 with b 6= 0 so that the Gaussian
curvature is zero, but the fibres are not lined up with the generators of the cylinder, then
the material will be subject to small shear as it falls over the cylinder.

The canonical form for an origami surface with zero Gaussian curvature (for example a
cone) can be written locally as

z =
√
ε(kx+ cy2) (40)

and if we lay the fabric over the surface with a fibre along the x axis, an analysis similar
to that used above for the convex surface leads to the shear being given by sin θ = 2εckv.
Thus we see that the only surface which can be covered with no shear at all is a cylinder
with one family of fibres lined up along its generators.

7.2 A nearly straight wing edge

We now consider the problem of covering the edge of a wing which was described in Sec-
tion 3. We take the surface of the wing to be

z = 0 if x < εf(y)

z = −m(x− εf(y)) if x > εf(y). (41)

where again ε is a small parameter. This takes the spar to be a plane of slope m but by
letting m → ∞ we will be able to allow it to be perpendicular to the wing surface. We
assume that when ε = 0 the fabric is spread over the two plane surfaces without any shear
and oriented with one of the fibre directions along the y-axis. Then we can write

x = u+ εx̂, y = v + εŷ, if x < εf(y)

x =
u√

1 +m2
+ εx̂, y = v + εŷ, if x > εf(y) (42)

and take as boundary conditions x = −1 on u = −1 and y = 0 on v = 0. If we substitute
these expressions into (34) and make x and y continuous across z = −m(x− εf(y)), then the
solution, correct to the first order in ε, is only affected in x > εf(y) where it is

x =
u√

1 +m2
+ εf(v)(1− 1√

1 +m2
)

y = v.

From this solution we find that the shear angle is given by

sin(θ) = ε|f ′(y)|[1− 1√
1 +m2

]. (43)

Taking the limit m → ∞, which corresponds to the two original planes being perpendicular,
the shear angle is given by ε|f ′(y)| as found in Section 4. Thus, to avoid a shear of more than
θmax, we need to make sure that the perturbation to the wing edge is such that ε|f ′(y)| <
sin(θmax) for all y.

7.3 Covering the corner of a cube

There was some interest in whether it was possible to cover the corner of a cube without
any wrinkling of the material. Using this model and assuming there is uniform shear in each
quadrant of the uv − plane it is possible to cover any pyramid by putting the origin at the
vertex of the pyramid. The total shear needed to cover the corner of a cube is π/2 since
it is neccessary to reduce the angle 2π on the flat unsheared surface to cover the 3 right
angles at the corner. This means that the angle between the two families of fibres in each
quadrant need to be sheared by π/8.

8 Supporting materials

A ZIP file containing the source code for the numerical simulations performed is embedded
within this PDF file. See the attachments tab within your PDF viewer to extract it. The source
code is compatible with Julia v1.x, an open source programming language for technical
computing.

esgi-airbus-code/esgi-meshgeneration.jl

# This file contains the code to generate meshes on arbitrary surfaces, e.g.,
# wing-box-like structures.

# It runs in Julia v1.0 (download from [https://julialang.org/downloads/])

# Put this in a folder alongside Project.toml and Manifest.toml and run Julia
# from that folder.

# At the Julia REPL type "]" to get into package management mode then type
#   activate .
#   instantiate
# to get get all the required packages. Press backspace to get out of package
# management mode.

# The code can then be loaded with
#   include("esgi-meshgeneration.jl")
# and run with
#   dostuff()

# This code requires a copy of MATLAB for the plotting routines; they can be
# replaced with other plotting routines as desired.

# The structure of the code is as follows.

#   * The top surface is defined by the function Z(y) and can be specified with
#     any of the parameteric forms provided (e.g., quadratic). You can use your
#     own form as desired.
#       * If you provide your own form, its derivative must also be provided
#         with the definition of a derivative function (cf.
#         derivative(::typeof(quadratic), y, λ)).
#   * The side surface is defined by the function X(y) in the same way as the top
#     surface.
#   * The function initiallattice provides the boundary conditions for the
#     computation by generating two fibres at 90° of each other (the angle is
#     arbitrary and entirely determined by this function).
#   * Computed points are stored as an (x, y, z) triple within a matrix that
#     represents the complete mesh.
#   * The function propagate! does the bulk of the computation by extending the
#     initial lattice according to the rules that fibres are inextensible.
#   * The function analyseangles is a helper function to calculate the maximum
#     shear angle. This function assumes that the fibres are initially at 90° to
#     each other.

using MATLABPlots
using StaticArrays
using Optim
using NLsolve
using QuadGK

const Point3d = SVector{3, Float64}

# Overload the plot function to plot the resulting fibre mesh
function MATLABPlots.plot(pts::Matrix{Point3d}; clear=true)
    if clear
        MATLABPlots.clf()
    end
    for i = 1:size(pts, 2)
        x = vec([pt[1] for pt in pts[:, i]])
        y = vec([pt[2] for pt in pts[:, i]])
        z = vec([pt[3] for pt in pts[:, i]])
        MATLABPlots.plot!(x, y, z, "k-")
    end
    for j = 1:size(pts, 1)
        x = vec([pt[1] for pt in pts[j, :]])
        y = vec([pt[2] for pt in pts[j, :]])
        z = vec([pt[3] for pt in pts[j, :]])
        MATLABPlots.plot!(x, y, z, "k-")
    end
    x = vec([pt[1] for pt in pts])
    y = vec([pt[2] for pt in pts])
    z = vec([pt[3] for pt in pts])
    MATLABPlots.plot!(x, y, z, ".")
    ymin = minimum(y)
    ymax = maximum(y)
    bpts = boundary.(range(ymin, stop=ymax, length=101))
    x = vec([pt[1] for pt in bpts])
    y = vec([pt[2] for pt in bpts])
    z = vec([pt[3] for pt in bpts])
    MATLABPlots.plot!(x, y, z, "r-")
end

#---
function X end  # placeholder
function Z end  # placeholder

# Distance calculations
arclength_integrand(f, y, λ) = sqrt(1 + derivative(f, y, λ)^2)
arclength(f, y₁, y₂, λ) = quadgk(y -> arclength_integrand(f, y, λ), y₁, y₂)[1]
distance(::typeof(X), f, pt1, pt2, λ) = sqrt(arclength(f, pt1[2], pt2[2], λ)^2 + (pt2[3] - pt1[3]).^2)
distance(::typeof(Z), f, pt1, pt2, λ) = sqrt(arclength(f, pt1[2], pt2[2], λ)^2 + (pt2[1] - pt1[1]).^2)

# Possible surfaces to use
flat(y, λ) = 0.0
arclength(::typeof(flat), y₁, y₂, λ) = y₂ - y₁

symquadratic(y, λ) = λ[1]*(1 - y^2)
arclength(::typeof(symquadratic), y₁, y₂, λ) = (2λ[1]*y₂*sqrt(4λ[1]^2*y₂^2+1) + asinh(2λ[1]*y₂))/(4λ[1]) - (2λ[1]*y₁*sqrt(4λ[1]^2*y₁^2+1) + asinh(2λ[1]*y₁))/(4λ[1])

quadratic(y, λ) = λ[1] + λ[2]*y + λ[3]*y^2
derivative(::typeof(quadratic), y, λ) = λ[2] + 2λ[3]*y

cubic(y, λ) = λ[1] + λ[2]*y + λ[3]*y^2 + λ[4]*y^3
derivative(::typeof(cubic), y, λ) = λ[2] + 2λ[3]*y + 3λ[4]*y^2

# Parameters for the surfaces in use
mutable struct Par
    X::Vector{Float64}
    Z::Vector{Float64}
end

const p = Par([], [])

p.Z = [0, 0, -0.2]  # top surface
p.X = [0, 0, 0.2]  # bottom surface

# Define the surfaces actually in use

# The top surface
Z(y) = quadratic(y, p.Z)
distance(::typeof(Z), pt1, pt2) = distance(Z, quadratic, pt1, pt2, p.Z) # this must match the line above in terms of the function used

# The side surface
X(y) = quadratic(y, p.X)
distance(::typeof(X), pt1, pt2) = distance(X, quadratic, pt1, pt2, p.X) # this must match the line above in terms of the function used

# The boundary between the two surfaces as a function of y
boundary(y) = SVector(X(y), y, Z(y))

onX(pt) = pt[1] - X(pt[2])
onZ(pt) = pt[3] - Z(pt[2])

isonX(pt) = isapprox(onX(pt), 0.0, rtol=1e-6)
isonZ(pt) = isapprox(onZ(pt), 0.0, rtol=1e-6)

# Calculate the distance between two points, irrespective of which surfaces they are on
function distance(pt1, pt2)
    pt1X = isonX(pt1)
    pt1Z = isonZ(pt1)
    pt2X = isonX(pt2)
    pt2Z = isonZ(pt2)
    if !(pt1X || pt1Z)
        error("pt1 is not on either surface")
    end
    if !(pt2X || pt2Z)
        error("pt1 is not on either surface")
    end
    if pt1X && pt2X
        return distance(X, pt1, pt2)
    elseif pt1Z && pt2Z
        return distance(Z, pt1, pt2)
    elseif pt1X && pt2Z
        return distance(X, pt1, Z, pt2)
    else
        return distance(Z, pt1, X, pt2)
    end
end

# Distance is commutative
distance(::typeof(Z), pt1, ::typeof(X), pt2) = distance(X, pt2, Z, pt1)

# Find the geodesic between two points on different surfaces - this requires an optimization routine for the general case
function distance(::typeof(X), pt1, ::typeof(Z), pt2)
    y0 = 0.5*(pt1[2] + pt2[2]) # mean as an initial guess
    f = y -> distance(X, pt1, boundary(y[1])) + distance(Z, boundary(y[1]), pt2)
    res = optimize(f, [y0], LBFGS())
    Optim.minimum(res)
end

# Generate two perpendicular fibres to start the computation; Δ is the distance between fibres and n specifies the number of fibres
function initiallattice(pt0, Δ::Number, n::Tuple{Int, Int})
    pt = Point3d(pt0[1], pt0[2], pt0[3])
    if !isonZ(pt)
        error("Starting point is not on Z")
    end
    pts = Matrix{Point3d}(undef, n[1], n[2])
    pts[1,1] = pt
    for i in 2:n[2]
        let pt = pts[1, i-1], Δ = Δ
            f = y -> distance(Z, pt, Point3d(pt[1], y[1], Z(y[1]))) - Δ
            res = nlsolve(f, [pt[2] + Δ], inplace=false)
            pts[1, i] = Point3d(pt[1], res.zero[1], Z(res.zero[1]))
        end
    end
    δ = Point3d(Δ, 0, 0)
    mirrored = false
    for j in 2:n[1]
        pts[j, 1] = pts[j-1, 1] + δ # mirror once past the boundary...
        if !mirrored && (onX(pts[j, 1]) > 0)
            mirrored = true
            d = distance(Z, boundary(pts[j, 1][2]), pts[j, 1])
            pts[j, 1] = boundary(pts[j, 1][2]) + SVector(0, 0, -d)
            δ = Point3d(0, 0, -Δ)
        end
    end
    pts
end

function projectto(::typeof(Z), pt)
    if isonZ(pt)
        return pt
    end
    bpt = boundary(pt[2])
    d = distance(X, bpt, pt)
    bpt + Point3d(d, 0, 0)
end

function projectto(::typeof(X), pt)
    if isonX(pt)
        return pt
    end
    bpt = boundary(pt[2])
    d = distance(Z, bpt, pt)
    bpt + Point3d(0, 0, -d)
end

function zeroproblem!(res, xy, pt1, pt2, δ)
    pt = Point3d(xy[1], xy[2], Z(xy[2]))
    if onX(pt) > 0
        pt = projectto(X, pt)
    end
    res[1] = distance(pt1, pt) - δ
    res[2] = distance(pt2, pt) - δ
end

# Calculate the location of the fibres based on the initial positions of two fibres
function propagate!(pts, Δ::Number)
    for i = 2:size(pts, 2)
        for j = 2:size(pts, 1)
            pt00 = projectto(Z, pts[j-1, i-1])
            pt10 = projectto(Z, pts[j, i-1])
            pt01 = projectto(Z, pts[j-1, i])
            v = pt10 - pt01
            midpt = pt01 + 0.5*v
            newpt = pt00 + 2*(midpt - pt00)
            res = nlsolve((res, xy) -> zeroproblem!(res, xy, pts[j, i-1], pts[j-1, i], Δ), [newpt[1], newpt[2]])
            pt = Point3d(res.zero[1], res.zero[2], Z(res.zero[2]))
            if onX(pt) > 0
                pt = projectto(X, pt)
            end
            pts[j, i] = pt
        end
    end
    pts
end

# Calculate the shear angles on each segment; this ignores segments with two vertices on one surface and two vertices on the other
function shearangles(pts)
    angles = zeros(size(pts, 1) - 1, size(pts, 2) - 1)
    for i = 2:size(pts, 2)
        for j = 2:size(pts, 1)
            ptlist = [pts[j-1, i-1], pts[j, i-1], pts[j, i], pts[j-1, i]]
            pton = [isonZ(pt) for pt in ptlist]
            if sum(pton) != 2 # not partially over the edge -> reliable angle
                for k = 1:4
                    if all(pton[1:3]) || !any(pton[1:3])
                        v1 = ptlist[1] - ptlist[2]
                        v2 = ptlist[3] - ptlist[2]
                        v1 /= norm(v1)
                        v2 /= norm(v2)
                        angles[j-1, i-1] = acos(dot(v1, v2))-π/2
                        continue
                    else
                        ptlist = circshift(ptlist, 1)
                    end
                end
            end
        end
    end
    angles
end

# Find how the maximum shear angle varies with a change in one of the parameters
function analyseangles(κ; Δ=0.05, n=(41, 21))
    p.Z = [0.0, 0.0, -κ]
    λ = range(κ-0.1, stop=κ+0.1, length=21)
    θ = zeros(length(λ))
    for i = 1:length(λ)
        p.X[3] = λ[i]
        pts = dostuff(Δ=Δ, n=n, doplot=false)
        θ[i] = maximum(abs.(shearangles(pts)))
    end
    (λ, θ)
end

# Give an example
function dostuff(; Δ = 0.1, n = (21, 11), doplot=true)
    pts = initiallattice(Point3d(-1, 0, Z(0)), Δ, n)
    propagate!(pts, Δ)
    if doplot
        plot(pts)
    end
    pts
end





